query: SELECT DISTINCT ?file ?filelink
  WITH
  {
    SELECT ?file
    WHERE
    {
      SERVICE wikibase:mwapi
      {
        bd:serviceParam wikibase:api "Generator" .
        bd:serviceParam wikibase:endpoint "commons.wikimedia.org" .
        bd:serviceParam mwapi:gcmtitle "Category:Photographs by Kevin Thiele" .
        bd:serviceParam mwapi:generator "categorymembers" .
        bd:serviceParam mwapi:gcmtype "file" .
        bd:serviceParam mwapi:gcmlimit "max" .
        bd:serviceParam mwapi:prop "globalusage" .
        bd:serviceParam mwapi:gulimit "max" .
        ?file wikibase:apiOutput mwapi:title .
        ?use wikibase:apiOutput "globalusage/gu/@wiki" .
      }
      FILTER BOUND(?use)
    }
  } AS %used_files
  WITH
  {
    SELECT ?file
    WHERE
    {
      SERVICE wikibase:mwapi
      {
        bd:serviceParam wikibase:api "Generator" .
        bd:serviceParam wikibase:endpoint "commons.wikimedia.org" .
        bd:serviceParam mwapi:gcmtitle "Category:Photographs by Kevin Thiele" .
        bd:serviceParam mwapi:generator "categorymembers" .
        bd:serviceParam mwapi:gcmtype "file" .
        bd:serviceParam mwapi:gcmlimit "max" .
        ?file wikibase:apiOutput mwapi:title .
      }
    }
  } AS %all_files
  WHERE
  {
    INCLUDE %all_files
    MINUS { INCLUDE %used_files }
    BIND (IRI(CONCAT("https://commons.wikimedia.org/wiki/", ENCODE_FOR_URI(?file))) AS ?filelink)
  }